Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematical Sciences
goutzicj@oneonta.edu

Lesson 8
Volumes of Solids of Revolution

Initializations

```
> restart;
  with(plots):
>
```

8.1 The Disk Method

The disk method is described by

$$dV = \pi\, r^2\, dx$$

where r is the radius of the disk, and dx denotes the thickness of the disk. The disk method always uses slices that are perpendicular to the axis of revolution.

Mathematical details will be provided in class.

Examples

Example 8.1.1

Compute the volume of the solid of revolution obtained by revolving the curve

$$f(x) = x\sqrt{4-x^2} \qquad 0 \le x \le 2$$

about the *x*-axis.

Solution
First make a sketch.

```
> f:=x->x*sqrt(4-x^2);
  plot(f(x), x=0..2);
```

$$f := x \to x\sqrt{4-x^2}$$



The volume of the disk created by taking a thin vertical slice is given by $dV = \pi f^2(x)\ dx$. Integration yields

$$V = \pi \int_0^2 f^2(x)\ dx$$

```
> e1:=Pi*Int(f(x)^2, x=0..2);
```

$$e1 := \pi \left(\int_0^2 x^2 \left(4 - x^2\right) \mathrm{d}x \right) \qquad \textbf{(2.1.1.1)}$$

```
> volume:=value(e1);
  evalf(volume);
```

$$volume := \frac{64}{15}\pi$$

$$13.40412866 \qquad \textbf{(2.1.1.2)}$$

It is possible to create the resulting solid of revolution. In Calculus III you will learn the mathematics behind this code.

```
> plot3d([x, f(x)*cos(u), f(x)*sin(u)], x=0..2, u=0..2*Pi,
  style=patch, axes=boxed, labels=[x,y,z], grid=[30,30],
  orientation=[-68, 62], shading=ZHUE);
```




```
>
```

8.2 The method of Cylindrical Shells

The method of cylindrical shells can be expressed as

$$dV = 2\,\pi\, r\, L\, dx$$

where *r* is the radius of the shell, *L* is the length of the shell, and *dx* denotes the thickness of the shell. The cylindrical shell method always uses slices that are parallel to the axis of revolution.

Mathematical details will be provided in class.

Examples

Example 8.2.2

Compute the volume of the solid of revolution obtained by revolving the curve

$$f(x) = \frac{1}{4}\, x\, \sqrt{27 - x^3} \qquad 0 \le x \le 3$$

about the *y*-axis.

Solution

In some cases it is advantageous to choose slices which are parallel to the axis of revolution. This problem is an example of that situation.

```
> f:=x->x*sqrt(27-x^3)/4;
  plot(f(x), x=0..3);
```

$$f := x \rightarrow \frac{1}{4}\, x\, \sqrt{27 - x^3}$$



Observe that using horizontal slices would require us to solve the non-trivial equation

$$y = x\sqrt{27 - x^3}$$

for the variable x. To avoid this difficulty we will work with vertical slices (parallel to the axis of revolution). If one of these slices is revolved about the *y*-axis, a so-called cylindrical shell results with volume

$$dV = 2\,\pi\, x\, f(x)\; dx$$

Subsequently, the volume of the created solid is given by

$$.V = 2\,\pi \int_0^3 x\, f(x)\; dx$$

```
> e1:=2*Pi*Int(x*f(x), x=0..3);
```

$$e1 := 2\,\pi \left(\int_0^3 \frac{1}{4}\, x^2 \sqrt{27 - x^3}\; \mathrm{d}x \right) \qquad \textbf{(3.1.1.1)}$$

```
> volume:=value(e1);
  evalf(volume);
```

$$volume := 9\,\pi\sqrt{3}$$

$$48.97258286 \qquad \textbf{(3.1.1.2)}$$

Of course the resulting solid can once again be visualized.

```
> plot3d([x*cos(u), x*sin(u), f(x)], x=0..3, u=0..2*Pi,
  style=patch, axes=boxed, labels=[x,y,z], scaling=
  constrained, shading=ZHUE, grid=[25, 45]);
```


-3
2
-3
z
0
y
1
1
x

>